
% Dynamic Macroeconomics
% Chapter 3
%
% Solow Model in discrete time
% _____________________________________
%
% Three permanent shocks are introduced in order to study 
% the dynamic behavior of the model
%
% 1. A 1 per cent shock to the savings rate (endval of x 1.01) 
% 2. A 1 per cent shock to total factor productivity (endval of a 
1.01)
% 3. A 1 per cent shock to population growth (endval of z 1.01)
% _____________________________________
%
% Endogenous Variables: 
% y output, k capital, c consumption
% w real wage, r real interest rate, sav savings rate
%
% Exogenous Variables and Shocks
% a total factor productivity, x shock to the savings rate
% z shock to population growth
%
% Parameters
% alpha share of capital in production, delta depreciation rate
% n population growth rate, g rate of technical progress
% s initial savings rate

var y k c w r sav;
varexo a x z;

parameters alpha delta n g s;

alpha=0.333;
delta=0.03;
n=0.01;
g=0.02;
s=0.30;

model;

y=a*(k(-1)^alpha);
c=(1-(s*x))*y;
k=(1/((1+n*z)*(1+g)))*(((s*x)*y)+(1-delta)*k(-1));
r=(alpha*a*k(-1)^(alpha-1))-delta;
w=(1-alpha)*a*k(-1)^(alpha);
sav=s*x;

end;

initval;

k=5.3;



c=1.38;
y=1.7;
a=1;
r=0.075;
w=1.16;
x=1;
z=1;

end;

steady;

endval;

k=5.3;
c=1.38;
y=1.7;
a=1.0;
r=0.075;
w=1.16;
x=1.0;
z=1.01;

end;

steady;
check;

simul(periods=100);

% Plotting Capital Output Consumption Real Interest Rate and the 
Real Wage

subplot(3,2,1); plot(k(1:100,1)); title('Capital');
subplot(3,2,2); plot(y(1:100,1)); title('Output');
subplot(3,2,3); plot(c(1:100,1)); title('Consumption');
subplot(3,2,4); plot(r(1:100,1)); title('Real Interest Rate');
subplot(3,2,5); plot(w(1:100,1)); title('Real Wage');
subplot(3,2,6); plot(sav(1:100,1)); title('Savings Rate');


